HEWLETT-PACKARD

Low-Cost Production Testing and Instrument Control

Summary

All products need to be tested during the production process. However, much production testing and calibration is still being done by technicians using stand-alone instruments.

Hewlett-Packard's test systems – including dedicated PC board testers, HP 1000 and Series 300 controllers, and a wide variety of HP-IB instruments – have been in the medium-to-high performance and price range. Now, the HP-71B offers you a low-to-medium performance and price alternative for adding intelligent instrument control to many of your manufacturing test stations.

Automated test stations provide higher throughput and greater accuracy and consistency than manual methods, and you will need fewer technicians or highly-trained workers. Adding low-cost HP-71B/HP-IL computerized testing to your manufacturing test stations will be highly profitable in most cases.

HP-IL test/monitoring systems controlled by HP-71B computers take up little bench space because HP-IL instruments are generally smaller (and less expensive) than their HP-IB counterparts, and the HP-71B is small enough to be handheld. HP-IL system performance is in the medium range, making the HP-71B/HP-IL combination an excellent choice for low-cost computerized test stations where space is limited and high-speed processing is not critical.

For additional information on networking HP-71Bs into supervisory systems, refer to HP application note 5954-1312. For information about using the HP-71B to monitor your production line, refer to HP application note 5954-1316.

*Third-party RAM Modules are available to expand RAM to 128K bytes.

Burn-In Testing Example

Burn-in testing is an application that is based on the monitoring capability of the HP-71/HP 3421A combination and the HP 82479A Data Acquisition Pac. After your product is manufactured, it should be given a burn-in run to eliminate infant-mortality failures. This helps to ensure the product will perform properly during, and for a reasonable time beyond, its warranty period.

The HP 3421A Data Acquisition/ Control Unit is connected to several test points on the new unit so its performance can be continuously displayed on a computer monitor. If any of the specified set points, such as current or temperature, is exceeded during burn-in, the HP-71 will perform an appropriate limit-action function.

Graphs of performance versus time are probably the most useful types of displays for hands-on manufacturing use. For example, a graph showing several test parameters can visually flag the parameter values existing at the time of a failure. The addition of an HP 2225B HP-IL ThinkJet Printer, controlled by your test program, to your production-monitoring station provides for a permanent visual record to help you modify your process and improve your manufacturing productivity. For information about printer control programming, refer to your ThinkJet Printer Owner's Manual.

Loaded PC Board Testing Example

Another application for an HP-71B/HP-IL data acquisition/control system is testing loaded PC boards. Tests are performed on the boards as they move along the assembly line, and test results such as *good, failed* and *cause of failure* are then reported on the HP-71B display, a printer or a computer monitor.

In this example, a keyboard printed circuit assembly is tested for:

- Idle current (where the keyboard's CMOS RAM is maintained while waiting for a key to be pressed)
- Oscillator frequency
- Simulated keystrokes
- Active current (where the operations associated with a keystroke are performed)

Because the keyboard is battery powered, it must be tested at both the high and low battery voltage limits, for a total of eight tests. To maximize battery life, the keyboard must not draw excessive current during operation.

At this point, the "keyboard" is a loaded PC board containing electronic components and ICs, but without buttons, switches, or power supply. To test keyboard functionality, the test fixture must supply both low and high voltage power and make the electrical connections to the HP 3421A Data Acquisition/Control Unit required to measure current and frequency and simulate keystrokes. The program in the HP-71B instructs the HP 3421A to take the measurements, switch the power supply voltage level and simulate a keystroke.

The following list of equipment required for the example will help you determine what equipment you need for your own production test applications:

- HP-71B Handheld Computer
- HP 82401A HP-IL Module
- HP 82420A RAM Module*
- HP 82167A, B, or D HP-IL Cables (2)
- HP 3421A Data Acquisition/ Control Unit
- HP 82479A Data Acquisition Pac

Optional:

- HP 9114A HP-IL Disc Drive (to store the data)
- HP 2225B HP-IL Think Jet Printer (to print the results)

Operation

To start a test, the operator presses a button on the test fixture that is connected to a digital I/O line in the HP 3421A Data Acquisition/Control Unit. This tells the unit to send a service request to the HP-71B which then runs the test program. The program controls the test's progress while monitoring and displaying the test results. These results can be sent to a printer for a hard copy record, and to a supervisory computer for use by management if a networking function is provided.

In the test subroutine, the HP-71B sets the input voltage to low by closing the proper HP 3421A actuator. The HP 3421A then measures input voltage, current and oscillator frequency. A key press is simulated by closing and then opening the proper digital I/O line in the HP 3421A, using a momentary-contact button. The power supply is then switched to high voltage and the tests repeated.

Program Listings

The following program listings illustrate how the HP-71B/ HP 3421A combination can be used in production testing. The first listing illustrates production testing using the additional BASIC keywords provided by the HP 82479A Data Acquisition Pac. The second listing is the same program using only the keywords provided by the HP 82401A HP-IL Interface. The Data Acquisition Pac makes the program run faster and allows more compact code. Your specific application will dictate which program is more suitable.

This code is written as an illustrative example. The code is linear and written to be as clear and straightforward as possible. However, it could be shortened considerably by using the advanced programming features of the HP-71B.

Program One

Line # Program Statement

```
1000 ! This code illustrates how one could use the HP-71 and the
1010 ! HP 82479A Data Acquisition Pac's keywords to control the
1020 ! HP 3421A doing product testing on a manufacturing line.
1030 !
1040 !        Variable Usage:
1050 ! V1 = LOW input voltage
1060 ! V3 = HIGH input voltage
1070 ! C1 = Current at LOW voltage in IDLE state
1080 ! C2 = Current at LOW voltage in ACTIVE state
1090 ! C3 = Current at HIGH voltage in IDLE state
1100 ! C4 = Current at HIGH voltage in ACTIVE state
1110 ! F1 = Oscillator frequency at LOW voltage in IDLE state
1120 ! F2 = Oscillator frequency at LOW voltage in ACTIVE state
1130 ! F3 = Oscillator frequency at HIGH voltage in IDLE state
1140 ! F4 = Oscillator frequency at HIGH voltage in ACTIVE state
1150 !
1160 !        It all starts here:
1170 'START':
1180 REAL V1,V3,C1,C2,C3,C4,F1,F2,F3,F4
1190 DIM K$[3]
1200 USER OFF
1210 INIT3421
1220 !        Main loop starts here:
1230 'LOOP':
1240 DISP "READY FOR NEXT TEST"
1250 ENABLE INTR 8 ! Interrupt on service request
1260 ON INTR GOTO 'SRQ' ! When interrupt occurs goto SRQ
1270 AUTOIDY ON ! Continuously send IDYs to see the service request
1280 SETSRQ 8 ! Enable the HP 3421A to send a service request
1290 MONITOR HIGH 20 ! Wait for operator to press start button
1300 ! Closes digital I/O line 20, service request sent
1310 K$=KEY$ ! Check key buffer
1320 IF K$#"" THEN 'MAYBE' ! If a key is in the buffer,
1330 ! see if we're supposed to quit
1340 GOTO 1310
1350 !
1360 'SRQ': OFF INTR ! Don't allow any more interrupts
1370 AUTOIDY OFF ! Stop continuously sending IDYs
1380 IF NOT BIT(READINTR,3) THEN GOTO 'LOOP'
1390 !
```

Continued

Line #	Program Statement
1400 !	Begin tests at LOW voltage:
1410 CLOSE 0 ! Switch on low power	
1420 V1=DCVOLTS(3) ! Input voltage	
1430 C1=DCVOLTS(4)/50*1000 ! IDLE current mA,	
1440 ! assuming 50Ω resistor	
1450 F1=FREQ(5)/1000 ! Oscillator frequency, kHz	
1460 CLOSE 23 ! Closes channel 23; start the key press	
1470 OPEN 23 ! Opens channel 23; completes key press	
1480 C2=DCVOLTS(4)/50*1000 ! ACTIVE current, mA	
1490 F2=FREQ(5)/1000 ! Oscillator frequency, kHz	
1500 OPEN 0 ! Turn off LOW voltage power	
1510 !	
1520 !	Begin tests at HIGH voltage:
1530 CLOSE 1 ! Switch on HIGH voltage power	
1540 V3=DCVOLTS(3) ! Input voltage	
1550 C3=DCVOLTS(4)/50*1000 ! IDLE current, mA	
1560 F3=FREQ(5)/1000 ! Oscillator frequency, kHz	
1570 CLOSE 23 ! Simulate key press	
1580 OPEN 23 ! Release key press	
1590 C4=DCVOLTS(4)/50*1000 ! ACTIVE current, mA	
1600 F4=FREQ(5)/1000 ! Oscillator frequency, kHz	
1610 OPEN ! Opens all channels, everything is disconnected	
1620 !	
1630 !	Test for valid measurements:
1640 IF V1<4.4 OR V1>4.6 THEN PRINT "Error--LOW Voltage: ";V1; " Volts"	
1650 IF V3<6.0 OR V3>6.2 THEN PRINT "Error--HIGH Voltage: ";V3; " Volts"	
1660 IF C1>.07 THEN PRINT "Error--IDLE Current = ";C1;"mA at LOW volts"	
1670 IF C2>10 THEN PRINT "Error--ACTIVE Current = ";C2;"mA at LOW volts"	
1680 IF C3>.08 THEN PRINT "Error--IDLE Current = ";C3;"mA at HIGH volts"	
1690 IF C4>10 THEN PRINT "Error--ACTIVE Current = ";C4;"mA at HIGH volts"	
1700 IF F1#0 THEN PRINT "Error--Oscillator = ";F1;"kHz at LOW volts --IDLE"	
1710 IF F2<610 OR F2>660 THEN PRINT "Error--Oscillator = ";F2;"kHz at LOW volts--ACTIVE"	
1720 IF F3#0 THEN PRINT "Error--Oscillator = ";F3;"kHz at HIGH volts--IDLE"	
1730 IF F4<610 OR F4>660 THEN PRINT "Error--Oscillator = ";F4;"kHz at HIGH volts--ACTIVE"	

```
1740 !
1750 ! This routine reads the key buffer until it is empty
1760 ! It checks on each read for the "Q" key to indicate it is time to quit
1770 ! This allows the operator to accidentally press a wrong key
1780 ! before pressing the "Q" and still have the program exit.
1790 ! If only the first key press was checked, then a wrong key
1800 ! press would force the operator to go through another test
1810 ! sequence before exiting.
1820 'NEXT': K$=KEY$ ! Get key from key buffer
1830 'MAYBE': IF K$="Q" THEN 'QUIT'
1840 IF K$#"" THEN 'NEXT' ! Go get the next key until buffer is empty
1850 GOTO 'LOOP' ! When buffer is empty return to main loop
1860 !
1870 !        It all ends here:
1880 'QUIT':
1890 AUTOIDY OFF ! Turn off IDYs
1900 INIT3421 ! Return HP 3421 to default state
1910 END
```

Program Two

```
1000 ! This code illustrates how one could use the HP-71 to control
1010 ! the HP 3421A Data Acquisition Unit doing product testing on a
1020 ! manufacturing line.
1030 !
1040 !        Variable Usage:
1050 ! V1 = LOW input voltage
1060 ! V3 = HIGH input voltage
1070 ! C1 = Current at LOW voltage in IDLE state
1080 ! C2 = Current at LOW voltage in ACTIVE state
1090 ! C3 = Current at HIGH voltage in IDLE state
1100 ! C4 = Current at HIGH voltage in ACTIVE state
1110 ! F1 = Oscillator frequency at LOW voltage in IDLE state
1120 ! F2 = Oscillator frequency at LOW voltage in ACTIVE state
1130 ! F3 = Oscillator frequency at HIGH voltage in IDLE state
1140 ! F4 = Oscillator frequency at HIGH voltage in ACTIVE state
1150 !
1160 !        It all starts here:
1170 'START':
```

Continued

Line #	Program Statement
1180	REAL V1,V3,C1,C2,C3,C4,F1,F2,F3,F4
1190	INTEGER H
1200	DIM K$[3]
1210	USER OFF
1220	H=DEVADDR ("HP3421") ! Get address of HP 3421
1230	OUTPUT :H ;"RS" ! Reset HP 3421
1240	!
1250	! Main loop starts here:
1260	'LOOP':
1270	DISP "READY FOR NEXT TEST"
1280	ENABLE INTR 8 ! Interrupt on service request
1290	ON INTR GOTO 'SRQ' ! When interrupt occurs goto SRQ
1300	OUTPUT :H ;"M8" ! HP 3421 is to send service request on channel closure
1310	OUTPUT :H ;"MH20" ! Monitor channel 20 for when it goes high
1320	SEND IDY ! Wait for service request
1330	K$=KEY$! Check key buffer
1340	IF K$#"" THEN 'MAYBE' ! If a key is in the buffer
1350	! see if we're supposed to quit
1360	GOTO 1320
1370	'SRQ': OFF INTR ! Don't allow any more interrupts
1380	IF NOT BIT(READINTR,3) THEN GOTO 'LOOP' ! Check interrupt mask
1390	!
1400	! Begin tests at LOW voltage:
1410	OUTPUT :H ;"CLSO" ! Switch on LOW power
1420	OUTPUT :H ;"DCV3" ! Read volts on channel 3
1430	ENTER :H ;V1 ! Input voltage
1440	OUTPUT :H ;"DCV4" ! Read current on channel 4
1450	ENTER :H ;C1 @ C1=C1/50*1000
1460	! IDLE current mA, assuming a 50 ohm resistor
1470	OUTPUT :H ;"FRQ5" ! Read frequency on channel 5
1480	ENTER :H ;F1 @ F1=F1/1000 ! Oscillator frequency, kHz
1490	OUTPUT :H ;"CLS23" ! Closes channel 23; start the key press
1500	OUTPUT :H ;"OPN23" ! Opens channel 23; completes key press
1510	OUTPUT :H ;"DCV4" ! Read current on channel 4
1520	ENTER :H ;C2 @ C2=C2/50*1000 ! ACTIVE current, mA
1530	OUTPUT :H ;"FRQ5" ! Read frequency on channel 5
1540	ENTER :H ;F2 @ F2=F2/1000 ! Oscillator frequency, kHz
1550	OUTPUT :H ;"OPN0" ! Turn off LOW voltage power
1560	!

Line #	Program Statement

```
1570 !        Begin tests at HIGH voltage:
1580 OUTPUT :H ;"CLS1" ! Switch on HIGH voltage power
1590 OUTPUT :H ;"DCV3" ! Read volts on channel 3
1600 ENTER :H ;V3 ! Input voltage
1610 OUTPUT :H ;"DCV4" ! Read current on channel 4
1620 ENTER :H ;C3 @ C3=C3/50*1000 ! IDLE current, mA
1630 OUTPUT :H ;"FRQ5" ! Read frequency on channel 5
1640 ENTER :H ;F3 @ F3=F3/1000 ! Oscillator frequency, kHz
1650 OUTPUT :H ;"CLS23" ! Simulate key press
1660 OUTPUT :H ;"OPN23" ! Release key press
1670 OUTPUT :H ;"DCV4" ! Read current on channel 4
1680 ENTER :H ;C4 @ C4=C4/50*1000 ! Active current, mA
1690 OUTPUT :H ;"FRQ5" ! Read frequency on channel 5
1700 ENTER :H ;F4 @ F4=F4/1000 ! Oscillator frequency, kHz
1710 OUTPUT :H ;"OPN" ! Opens all channels, everything is disconnected
1720 !        Test for valid measurements:
1730 IF V1<4.4 OR V1>4.6 THEN PRINT "Error--LOW Voltage: ";V1;" Volts"
1740 IF V3<6.0 OR V3>6.2 THEN PRINT "Error--HIGH Voltage: ";V3;" Volts"
1750 IF C1>.07 THEN PRINT "Error--IDLE Current = ";C1;"mA at LOW Volts"
1760 IF C2>10 THEN PRINT "Error--ACTIVE Current = ";C2;"mA at LOW
     Volts"
1770 IF C3>.08 THEN PRINT "Error--IDLE Current = ";C3;"mA at HIGH
     Volts"
1780 IF C4>10 THEN PRINT "Error--ACTIVE Current = ";C4;"mA at HIGH
     Volts"
1790 IF F1#0 THEN PRINT "Error--Oscillator = ";F1;"kHz at LOW Volts--Idle"
1800 IF F2<610 OR F2>660 THEN PRINT "Error--Oscillator = ";F2;"kHz
     at LOW Volts--Active"
1810 IF F3#0 THEN PRINT "Error--Oscillator = ";F3;"kHz at HIGH Volts--Idle"
1820 IF F4<610 or F4>660 THEN PRINT "Error--Oscillator = ";F4;"kHz
     at HIGH Volts--Active"
1830 !
1840 ! This routine reads the key buffer until it is empty. It
1850 ! checks on each read for the "Q" key to indicate it is time to quit.
1860 ! This allows the operator to accidentally press a wrong key
1870 ! before pressing the "Q" and still have the program exit.
1880 ! If only the first key press was checked, then a wrong key
```

Continued

Line #	Program Statement
1890	! press would force the operator to go through another test
1900	! sequence before exiting.
1910	'NEXT': K$=KEY$! Get key from key buffer
1920	'MAYBE': IF K$="Q" THEN 'QUIT'
1930	IF K$#"" THEN 'NEXT' ! Go get the next key until buffer is empty
1940	GOTO 'LOOP' ! When buffer is empty return to main loop
1950	!
1960	! It all ends here:
1970	'QUIT':
1980	OUTPUT :H ;"RS" ! Reset the HP 3421, return it to default state
1990	END

For additional information about Hewlett-Packard's handheld computers in manufacturing solutions, call the HP office nearest you and ask for your technical computer or instrument representative. Local HP sales offices are listed in the white pages of your telephone book.

Hewlett-Packard
Handheld Computer and
Calculator Operation
1000 N.E. Circle Blvd.
Corvallis, Oregon 97330

Printed in USA. 10/85
5954-1317 WCP 5.0